/* Replication Do File for
"The Politics of Enmity: Outgroup Hostility and Support for the Big Lie" (Benegal, Hayes, and Ugues, 2026, in Political Research Quarterly)

This file contains commands in Stata for data cleaning, coding, and analysis to replicate the results presented in the manuscript and appendix.

The associated data file is "anes_timeseries_20251104_withids.dta".
This is the 2024 ANES survey data file of that date which includes merged responses for the white identity question items (not included in original ANES release).
*/

clear
set more off
// set working directory:
cd "C:\Users\salil\Dropbox\Big Lie Article Project\PRQ RR Jan 26\Post-acceptance docs\replication\"
use "anes_timeseries_20251104_withids.dta" 

// Independent and control variables

tab V241551, m
tab V241551 if V241551>0
gen gender=.
replace gender=0 if V241551==1
replace gender=1 if V241551==2
label define gender 0 "Men" 1 "Women"
label values gender gender
tab gender V241551, m
gen female=gender
gen male=.
replace male=0 if female==1
replace male=1 if female==0
label var female "Women"

tab V241501x
gen race=.
replace race=V241501x if !missing(V241501x) & V241501x>0
label define race 1 "White" 2 "Black" 3 "Hispanic" 4 "AAPI" 5 "Native American" 6 "Multiple/other"
label values race race

// Racial resentment
tab V242300
tab V242301
tab V242302
tab V242303
gen favors=.
replace favors=1 if V242300==5
replace favors=2 if V242300==4
replace favors=3 if V242300==3
replace favors=4 if V242300==2
replace favors=5 if V242300==1
gen generations=.
replace generations=1 if V242301==1
replace generations=2 if V242301==2
replace generations=3 if V242301==3
replace generations=4 if V242301==4
replace generations=5 if V242301==5
gen deserve=.
replace deserve=1 if V242302==1
replace deserve=2 if V242302==2
replace deserve=3 if V242302==3
replace deserve=4 if V242302==4
replace deserve=5 if V242302==5
gen tryhard=.
replace tryhard=1 if V242303==5
replace tryhard=2 if V242303==4
replace tryhard=3 if V242303==3
replace tryhard=4 if V242303==2
replace tryhard=5 if V242303==1

factor favors generations deserve tryhard
alpha favors generations deserve tryhard
gen resentment=(favors+generations+deserve+tryhard)/4
label var resentment "Racial resentment"

// Modern Sexism scale
tab V242280
tab V242281
gen equality=.
replace equality=1 if V242280==5
replace equality=2 if V242280==4
replace equality=3 if V242280==3
replace equality=4 if V242280==2
replace equality=5 if V242280==1
gen complain=.
replace complain=1 if V242281==5
replace complain=2 if V242281==4
replace complain=3 if V242281==3
replace complain=4 if V242281==2
replace complain=5 if V242281==1

factor equality complain
alpha equality complain

gen modernsexism=(equality+complain)/2
label var modernsexism "Modern sexism"

// Anti-immigrant sentiment

**V242453 POST: CSES6- Q05C: OUT-GROUP ATTITUDES: IMMIGRANTS INCREASE CRIME
gen immigrants_crime=.
replace immigrants_crime=1 if V242453==5
replace immigrants_crime=2 if V242453==4
replace immigrants_crime=3 if V242453==3
replace immigrants_crime=4 if V242453==2
replace immigrants_crime=5 if V242453==1
label var immigrants_crime "immigrants_crime"
label define immigrants_crime 1 "strongly disagree"  5 "strongly agree" 
label value immigrants_crime immigrants_crime
tab immigrants_crime

***V242236 POST: HOW IMPORTANT IS IT FOR IMMIGRANTS TO ADAPT TO CUSTOMS/TRADITIONS OF THE US
gen immigrants_customs=.
replace immigrants_customs=1 if V242236==1
replace immigrants_customs=2 if V242236==2
replace immigrants_customs=3 if V242236==3
replace immigrants_customs=4 if V242236==4
replace immigrants_customs=5 if V242236==5
label var immigrants_customs "immigrants_customs"
label define immigrants_customs 1 "not important at all"  5 "extremely important" 
label value immigrants_customs immigrants_customs
tab immigrants_customs

*V242235 POST: ARE IMMIGRANTS GENERALLY GOOD OR BAD FOR AMERICA'S ECONOMY
gen immigrants_economy=.
replace immigrants_economy=1 if V242235==1 
replace immigrants_economy=2 if V242235==2 | V242235==3
replace immigrants_economy=3 if V242235==4
replace immigrants_economy=4 if V242235==5 | V242235==6
replace immigrants_economy=5 if V242235==7
label var immigrants_economy "immigrants_economy"
label define immigrants_economy 1 "extremely good"  5 "extremely bad" 
label value immigrants_economy immigrants_economy
tab immigrants_economy

alpha immigrants_crime immigrants_customs immigrants_economy, gen(xenophobia)
label var xenophobia "Anti-immigrant sentiment"

gen age_cohort=.
replace age_cohort=1 if V241458x>=18 & V241458x<=25
replace age_cohort=2 if V241458x>=26 & V241458x<=40
replace age_cohort=3 if V241458x>=41 & V241458x<=55
replace age_cohort=4 if V241458x>=55 & V241458x<=65
replace age_cohort=5 if V241458x>65 
label var age_cohort "age_cohort"
label define age_cohort 1 "18-25" 2 "25-40" 3 "41-55" 4 "56-65" 5 "over 65" 
label value age_cohort age_cohort
tab age_cohort
tab age_cohort

// PartyID
tab V241227x
gen partyid7=.
replace partyid7=V241227x if V241227x>0 & !missing(V241227x)
label define partyid7 1 "Strong Democrat" 2 "Democrat" 3 "Ind-lean Democrat" 4 "Independent" 5 "Ind-lean Republican" 6 "Republican" 7 "Strong Republican"
label values partyid7 partyid7

gen partyid3=.
replace partyid3=1 if partyid7==1 | partyid7==2 | partyid7==3
replace partyid3=2 if partyid7==4
replace partyid3=3 if partyid7==5 | partyid7==6 | partyid7==7
label define partyid3 1 "Democrat" 2 "Independent" 3 "Republican"
label values partyid3 partyid3

tab partyid3, gen(p)
rename p1 Democrat
label var Democrat "Democratic"
rename p2 Independent
label var Independent "Independent"
rename p3 Republican
label var Republican "Republican"

tab V241465x
label define educ 1 "HS or less" 2 "Some college/no bachelor's" 3 "Bachelor's degree" 4 "Graduate degree"
gen educ=.
replace educ=1 if V241465x==1 | V241465x==2
replace educ=2 if V241465x==3
replace educ=3 if V241465x==4
replace educ=4 if V241465x==5
label values educ educ
tab educ V241465x, m

gen college=.
replace college=0 if educ==1 | educ==2
replace college=1 if educ==3 | educ==4
label define college 0 "No college degree" 1 "College degree or higher"
label values college college
tab college V241465x, m
label var college "College degree"

tab V241567x
gen income=.
replace income=V241567x if !missing(V241567x) & V241567x>0
label define income 1 "Under $10K" 2 "$10K to <$30K" 3 "$30K to <$60K" 4 "$60K to <100K" 5 "100K to <$250K" 6 "$250K+"
label values income income
label var income "Income"

tab V241177
gen ideology=.
replace ideology=V241177 if !missing(V241177) & V241177>0 & V241177<8
label define ideology  1 "Extremely liberal" 2 "Liberal" 3 "Slightly liberal" 4 "Moderate" 5 "Slightly conservative" 6 "Conservative" 7 "Extremely conservative"
label values ideology ideology
tab ideology V241177
label var ideology "Ideology"

gen trumptherm=.
replace trumptherm=V241157 if V241157>=0
sum trumptherm
label var trumptherm "FT Trump"

gen trumpvote=.
replace trumpvote=1 if V242067==2
replace trumpvote=0 if V242067==1 | V242067==3 | V242067==4 | V242067==5 | V242067==6
label var trumpvote "trumpvote"
label define trumpvote 0 "Other" 1 "Trump"  
label value trumpvote trumpvote
tab trumpvote

gen harrisvote=.
replace harrisvote=1 if V242067==1
replace harrisvote=0 if V242067==2 | V242067==3 | V242067==4 | V242067==5 | V242067==6
label var harrisvote "harrisvote"
label define harrisvote 0 "Other" 1 "Harris"  
label value harrisvote harrisvote 
tab harrisvote 

gen strongleader=.
replace strongleader=1 if V242411==5 
replace strongleader=2 if V242411==4 
replace strongleader=3 if V242411==3 
replace strongleader=4 if V242411==2 
replace strongleader=5 if V242411==1 
label define strongleader 1 "strongly disagree" 5 "strongly agree"
label values strongleader strongleader
tab strongleader
label var strongleader "Strong leader pref"

gen goptherm2=.
replace goptherm2=1 if V242433==1 
replace goptherm2=2 if V242433==2 
replace goptherm2=3 if V242433==3 
replace goptherm2=4 if V242433==4 
replace goptherm2=5 if V242433==5 
replace goptherm2=6 if V242433==6 
replace goptherm2=7 if V242433==7 
replace goptherm2=8 if V242433==8 
replace goptherm2=9 if V242433==9 
replace goptherm2=10 if V242433==10 
label define goptherm2 0 "strongly dislike" 10 "strongly like"
label values goptherm2 goptherm2 
tab goptherm2 

tab V242531
gen ytidentity=.
replace ytidentity=1 if V242531==5 & race==1
replace ytidentity=2 if V242531==4 & race==1
replace ytidentity=3 if V242531==3 & race==1
replace ytidentity=4 if V242531==2 & race==1
replace ytidentity=5 if V242531==1 & race==1
label var ytid "White identity"

tab V242519
gen minorityhiring=.
replace minorityhiring=1 if V242519==5 & race==1
replace minorityhiring=2 if V242519==4 & race==1
replace minorityhiring=3 if V242519==3 & race==1
replace minorityhiring=4 if V242519==2 & race==1
replace minorityhiring=5 if V242519==1 & race==1
label var minority "White grievance"

factor ytid minorityhiring
alpha ytid minorityhiring

gen religion=.
replace religion=1 if V241422==1
replace religion=2 if V241422==2
replace religion=3 if V241422==3
replace religion=4 if V241422==4
replace religion=5 if V241422==5
replace religion=6 if V241422==6
replace religion=7 if V241422==7
replace religion=8 if V241422==8
replace religion=9 if V241422==9
replace religion=10 if V241422==10
replace religion=11 if V241422==11
replace religion=12 if V241422==12
label var religion "religion"
label define religion 1 "protestant" 2 "roman catholic" 3 "Orthodox Christian (such as Greek or Russian Orthodox)" 4 "Latter-Day Saints (LDS)" 5 "Jewish" 6 "Muslim" 7 "Buddhist" 8 "Hindu" 9"Atheist" 10 "Agnostic" 11 "Something else" 12 "Nothing in particular"
label value religion religion
tab religion

gen bornagain=.
replace bornagain=1 if V241442==1
replace bornagain=0 if V241442==2
replace bornagain=0 if V241442==-1
label var bornagain "bornagain"
label define bornagain 1 "born again christian" 0 "not born again" 
label value bornagain bornagain
tab bornagain

// Dependent variables

// misinfo - belief that Trump won the 2020 election
gen misinfo20=.
label define misinfo20 0 "Biden (correct)" 1 "Trump (misinfo)" 
replace misinfo20=0 if V242567==2
replace misinfo20=1 if V242567==1
label values misinfo20 misinfo20
tab misinfo20 V242567

// overconfidence - belief that Trump won (wrong answer/election result distrust), *and* moderate confidence or higher
gen dk2020=.
replace dk2020=0 if V242567==2
replace dk2020=0 if V242567==1 & V242568==1
replace dk2020=0 if V242567==1 & V242568==2

replace dk2020=1 if V242567==1 & V242568==3
replace dk2020=1 if V242567==1 & V242568==4
replace dk2020=1 if V242567==1 & V242568==5

label define dk2020 0 "Biden OR Trump w/ low confidence" 1 "Trump with moderate+ confidence"
label values dk2020 dk2020

// misinfo - belief that Jan 6 was carried out by FBI agents posing as pro-Trump protestors
gen Jan6misinfo=.
label define Jan6misinfo 0 "pro-Trump protestors (correct)" 1 "FBI agents posing (misinfo)" 
replace Jan6misinfo=0 if V242571==1
replace Jan6misinfo=1 if V242571==2
label values Jan6misinfo Jan6misinfo
tab Jan6misinfo V242571

// overconfidence - belief that Jan 6 was FBI agents (wrong answer/election result distrust), *and* moderate confidence or higher
gen dkJan6=.
replace dkJan6=0 if V242571==1
replace dkJan6=0 if V242571==2 & V242572==1
replace dkJan6=0 if V242571==2 & V242572==2

replace dkJan6=1 if V242571==2 & V242572==3
replace dkJan6=1 if V242571==2 & V242572==4
replace dkJan6=1 if V242571==2 & V242572==5

label define dkJan6 0 "Pro-Trump OR FBI w/ low confidence" 1 "FBI with moderate+ confidence"
label values dkJan6 dkJan6

tab race
gen white=.
replace white=1 if race==1
replace white=0 if race!=1 & !missing(race)
tab race white, m

/// results begin here

*laptop 
cd "C:\Users\salil\Dropbox\Big Lie Article Project\PRQ RR Jan 26\Post-acceptance docs"

// Table 2 
eststo clear
eststo a1: logit misinfo20 resentment xeno modernsexism age female income college b1.partyid3 ideology strongleader trumptherm bornagain if race==1
eststo a2: logit misinfo20 resentment xeno ytid minority modernsexism age female income college b1.partyid3 ideology strongleader trumptherm bornagain if race==1
eststo b1: logit dk2020 resentment xeno modernsexism age female income college b1.partyid3 ideology strongleader trumptherm bornagain if race==1
eststo b2: logit dk2020 resentment xeno ytid minority modernsexism age female income college b1.partyid3 ideology strongleader trumptherm bornagain if race==1

# delimit ;
esttab using Table2.rtf, nogap se b b(%9.4f) starlevels(* .05 ** .01 *** .005) r2(%9.3f) 
	title("Table 2: Belief and overconfidence in the big lie" )
	  label nonumbers mtitles("Big lie" "Big lie + white id" "Overconfidence" "Overconfidence + white id")
	addnote("Logistic regression. 2024 ANES data")
	compress replace;
 #delimit cr

// Table 3
eststo clear
eststo a1: logit Jan6misinfo resentment xeno modernsexism age female income college b1.partyid3 ideology strongleader trumptherm bornagain if race==1
eststo a2: logit Jan6misinfo resentment xeno ytid minority modernsexism age female income college b1.partyid3 ideology strongleader trumptherm bornagain if race==1
eststo b1: logit dkJan6 resentment xeno modernsexism age female income college b1.partyid3 ideology strongleader trumptherm bornagain if race==1
eststo b2: logit dkJan6 resentment xeno ytid minority modernsexism age female income college b1.partyid3 ideology strongleader trumptherm bornagain if race==1

# delimit ;
esttab using Table3.rtf, nogap se b b(%9.4f) starlevels(* .05 ** .01 *** .005) r2(%9.3f) 
	title("Table 3: Belief and overconfidence in misinformation about January 6" )
	  label nonumbers mtitles("Big lie" "Big lie + white id" "Overconfidence" "Overconfidence + white id")
	addnote("Logistic regression. 2024 ANES data")
	compress replace;
 #delimit cr

// Robustness checks for Appendix using religion categorial var instead of bornagain

// Table 2 
eststo clear
eststo a1: logit misinfo20 resentment xeno modernsexism age female income college b1.partyid3 ideology strongleader trumptherm i.religion if race==1
eststo a2: logit misinfo20 resentment xeno ytid minority modernsexism age female income college b1.partyid3 ideology strongleader trumptherm i.religion if race==1
eststo b1: logit dk2020 resentment xeno modernsexism age female income college b1.partyid3 ideology strongleader trumptherm i.religion if race==1
eststo b2: logit dk2020 resentment xeno ytid minority modernsexism age female income college b1.partyid3 ideology strongleader trumptherm i.religion if race==1

# delimit ;
esttab using Table2b.rtf, nogap se b b(%9.4f) starlevels(* .05 ** .01 *** .005) r2(%9.3f) 
	title("Table 2: Belief and overconfidence in the big lie" )
	  label nonumbers mtitles("Big lie" "Big lie + white id" "Overconfidence" "Overconfidence + white id")
	addnote("Logistic regression. 2024 ANES data")
	compress replace;
 #delimit cr

// Table 3
eststo clear
eststo a1: logit Jan6misinfo resentment xeno modernsexism age female income college b1.partyid3 ideology strongleader trumptherm i.religion if race==1
eststo a2: logit Jan6misinfo resentment xeno ytid minority modernsexism age female income college b1.partyid3 ideology strongleader trumptherm i.religion if race==1
eststo b1: logit dkJan6 resentment xeno modernsexism age female income college b1.partyid3 ideology strongleader trumptherm i.religion if race==1
eststo b2: logit dkJan6 resentment xeno ytid minority modernsexism age female income college b1.partyid3 ideology strongleader trumptherm i.religion if race==1

# delimit ;
esttab using Table3b.rtf, nogap se b b(%9.4f) starlevels(* .05 ** .01 *** .005) r2(%9.3f) 
	title("Table 3: Belief and overconfidence in misinformation about January 6" )
	  label nonumbers mtitles("Big lie" "Big lie + white id" "Overconfidence" "Overconfidence + white id")
	addnote("Logistic regression. 2024 ANES data")
	compress replace;
 #delimit cr

// Predicted Probabilities, Fig 1

 eststo clear 
	logit dk2020 resentment xeno ytid minority modernsexism age female income college b1.partyid3 ideology strongleader trumptherm bornagain if partyid3==3
	eststo rr: margins, at(resentment=(1(.5)5))   atmeans post 
	logit dk2020 resentment xeno ytid minority modernsexism age female income college b1.partyid3 ideology strongleader trumptherm bornagain if partyid3==3
	eststo xeno: margins, at(xeno=(1(.5)5))   atmeans post 
	logit dk2020 resentment xeno ytid minority modernsexism age female income college b1.partyid3 ideology strongleader trumptherm bornagain  if partyid3==3
	eststo ytid: margins, at(ytid=(1(.5)5))   atmeans post 
	logit dk2020 resentment xeno ytid minority modernsexism age female income college b1.partyid3 ideology strongleader trumptherm bornagain  if partyid3==3
	eststo ytjobs: margins, at(minority=(1(.5)5))   atmeans post 
	logit dk2020 resentment xeno ytid minority modernsexism age female income college b1.partyid3 ideology strongleader trumptherm bornagain  if partyid3==3
	eststo sexism: margins, at(modernsexism=(1(.5)5))   atmeans post 
	
coefplot (rr, label (Racial resentment)), ///
	title("") ///
	xtitle ("Racial Resentment") ///
	xlab(1 "Low" 3 "Medium" 5 "High", labsize(small)) ///
	ylab(0 "0%" .25 "25%" .5 "50%" .75 "75%", labsize(small)) ///
	at recast(line) lwidth(*2) ciopts(recast(rline) lpattern(dash)) name(rr, replace)

coefplot (xeno, label (Anti-immigrant resentment)), ///
	title("") ///
	xtitle ("Immigrant resentment") ///
	xlab(1 "Low" 3 "Medium" 5 "High", labsize(small)) ///
	ylab(0 "0%" .25 "25%" .5 "50%" .75 "75%", labsize(small)) ///
	at recast(line) lwidth(*2) ciopts(recast(rline) lpattern(dash)) name(xeno, replace)
	
coefplot (sexism, label (Sexism)), ///
	title("") ///
	xtitle ("Sexism") ///
	xlab(1 "Low" 3 "Medium" 5 "High", labsize(small)) ///
	ylab(0 "0%" .25 "25%" .5 "50%" .75 "75%", labsize(small)) ///
	at recast(line) lwidth(*2) ciopts(recast(rline) lpattern(dash)) name(sexism, replace)
	
coefplot (ytid, label (White identity)), ///
	title("") ///
	xtitle ("White identity") ///
	xlab(1 "Low" 3 "Medium" 5 "High", labsize(small)) ///
	ylab(0 "0%" .25 "25%" .5 "50%" .75 "75%", labsize(small)) ///
	at recast(line) lwidth(*2) ciopts(recast(rline) lpattern(dash)) name(ytid, replace)

	coefplot (ytjobs, label (White jobs)), ///
	title("") ///
	xtitle ("White jobs being taken by minorities") ///
	xlab(1 "Low" 3 "Medium" 5 "High", labsize(small)) ///
	ylab(0 "0%" .25 "25%" .5 "50%" .75 "75%", labsize(small)) ///
	at recast(line) lwidth(*2) ciopts(recast(rline) lpattern(dash)) name(ytjobs, replace)
	
graph combine rr xeno sexism ytjobs, col(2) title("Overconfidence that Trump won the 2020 election (for white Republicans)")

// Predicted Probabilities, Fig 2
 eststo clear 
	logit dkJan6 resentment xeno ytid minority modernsexism age female income college b1.partyid3 ideology strongleader trumptherm bornagain if partyid3==3
	eststo rr: margins, at(resentment=(1(.5)5))   atmeans post 
	logit dkJan6 resentment xeno ytid minority modernsexism age female income college b1.partyid3 ideology strongleader trumptherm bornagain  if partyid3==3
	eststo xeno: margins, at(xeno=(1(.5)5))   atmeans post 
	logit dkJan6 resentment xeno ytid minority modernsexism age female income college b1.partyid3 ideology strongleader trumptherm bornagain  if partyid3==3
	eststo ytid: margins, at(ytid=(1(.5)5))   atmeans post 
	logit dkJan6 resentment xeno ytid minority modernsexism age female income college b1.partyid3 ideology strongleader trumptherm bornagain  if partyid3==3
	eststo ytjobs: margins, at(minority=(1(.5)5))   atmeans post 
	logit dkJan6 resentment xeno ytid minority modernsexism age female income college b1.partyid3 ideology strongleader trumptherm bornagain  if partyid3==3
	eststo sexism: margins, at(modernsexism=(1(.5)5))   atmeans post 
	
coefplot (rr, label (Racial resentment)), ///
	title("") ///
	xtitle ("Racial Resentment") ///
	xlab(1 "Low" 3 "Medium" 5 "High", labsize(small)) ///
	ylab(0 "0%" .25 "25%" .5 "50%", labsize(small)) ///
	at recast(line) lwidth(*2) ciopts(recast(rline) lpattern(dash)) name(rr, replace)

coefplot (xeno, label (Anti-immigrant resentment)), ///
	title("") ///
	xtitle ("Immigrant resentment") ///
	xlab(1 "Low" 3 "Medium" 5 "High", labsize(small)) ///
	ylab(0 "0%" .25 "25%" .5 "50%", labsize(small)) ///
	at recast(line) lwidth(*2) ciopts(recast(rline) lpattern(dash)) name(xeno, replace)
	
coefplot (ytid, label (White identity)), ///
	title("") ///
	xtitle ("White identity") ///
	xlab(1 "Low" 3 "Medium" 5 "High", labsize(small)) ///
	ylab(0 "0%" .25 "25%" .5 "50%", labsize(small)) ///
	at recast(line) lwidth(*2) ciopts(recast(rline) lpattern(dash)) name(ytid, replace)

coefplot (ytjobs, label (White jobs)), ///
	title("") ///
	xtitle ("White jobs being taken by minorities") ///
	xlab(1 "Low" 3 "Medium" 5 "High", labsize(small)) ///
	ylab(0 "0%" .25 "25%" .5 "50%", labsize(small)) ///
	at recast(line) lwidth(*2) ciopts(recast(rline) lpattern(dash)) name(ytjobs, replace)
	
coefplot (sexism, label (Sexism)), ///
	title("") ///
	xtitle ("Sexism") ///
	xlab(1 "Low" 3 "Medium" 5 "High", labsize(small)) ///
	ylab(0 "0%" .25 "25%" .5 "50%" .75 "75%", labsize(small)) ///
	at recast(line) lwidth(*2) ciopts(recast(rline) lpattern(dash)) name(sexism, replace)	

graph combine rr xeno sexism ytjobs, col(2) title("Overconfidence that FBI caused Jan 6 (for white Republicans)")


// Additional appendix figure for effects of white identity.

eststo clear 
	logit dk2020 resentment xeno ytid minority modernsexism age female income college b1.partyid3 ideology strongleader trumptherm bornagain if partyid3==3
	eststo ytid: margins, at(ytid=(1(.5)5))   atmeans post 
	logit dkJan6 resentment xeno ytid minority modernsexism age female income college b1.partyid3 ideology strongleader trumptherm bornagain if partyid3==3
	eststo ytid2: margins, at(ytid=(1(.5)5))   atmeans post 

coefplot (ytid, label (White identity)), ///
	title("Overconfidence in big lie") ///
	xtitle ("White identity") ///
	xlab(1 "Low" 3 "Medium" 5 "High", labsize(small)) ///
	ylab(0 "0%" .25 "25%" .5 "50%", labsize(small)) ///
	at recast(line) lwidth(*2) ciopts(recast(rline) lpattern(dash)) name(ytid, replace)
	
coefplot (ytid2, label (White identity)), ///
	title("Overconfidence that FBI caused Jan 6") ///
	xtitle ("White identity") ///
	xlab(1 "Low" 3 "Medium" 5 "High", labsize(small)) ///
	ylab(0 "0%" .25 "25%" .5 "50%", labsize(small)) ///
	at recast(line) lwidth(*2) ciopts(recast(rline) lpattern(dash)) name(ytid2, replace)
	
graph combine ytid ytid2, col(2) title("Effects of white identity")


